No, I copied the code from a magazine as a kid, I literally read the code, line by line, and typed it on my computer. It was maybe 100 or 200 lines of code of BASIC.

The game only took YES and NO as an answer (starting with `is it a living thing` I think). Then, if you had already played the game, it would read the next questions from a flat file that corresponds to your latest answer and moved on to narrow down the answer (or rather the last question in the tree).

If it ran out of questions to ask for (the last question in the tree has no answer or no next question) it would give up and ask for what you had in mind and the question that would differentiate what you had in mind from the latest possibility.

Then at the next play, you would find the same question if you went along the same path and now it would know what you were thinking about.

I don't know if q20 is better than that. It just reminded me of that game.

edit: see here https://www.atarimagazines.com/v4n12/Animal.html

The game's intelligence is rooted in a tree-like knowledge structure. Each point where the tree has "branches" (or nodes) consists of three bits of information-a question to ask, a YES branch, and a NO branch. The terminations (tips) of the branches consist of a single bit of information-in this case, an animal.
During play, the program branches through the tree from the bottom to the top, stopping at each node to ask a question and branching according to the user's response. When it reaches a termination, its guess is the animal it finds there.
When the computer guesses incorrectly, the tree "grows" by creating a new node at the termination. Remember that a node is formed of a question, a YES branch, and a NO branch. To create the new node, the program uses the question you provide. It then places the new animal you have given it at the termination of the YES branch. The animal that was at the old termination gets pushed up the NO branch. Thus, what was previously a single termination becomes a node with two terminations, and the tree grows a little taller.
To get Animal Intelligence to work properly in BASIC, a language which lacks list processing ability, I had to divide the knowledge base into three parts. The first and second are simulated string arrays-AX$ holds the animals, and QX$ holds the questions. The third part, TREE$, is the knowledge tree.
Though it is an ordinary string, TREE$ functions as a matrix of pointers. Every three bytes comprise a node or termination. The first byte of a node is a pointer to a question held in QX$. The second and third bytes point to the YES branch and NO branch respectively The first byte of a termination is a pointer to an animal held in AX$, while the second and third bytes are just dummies at the start. Later they will be used as pointers when the termination grows into a node.
Animal Intelligence isn't limited to animals, however. With some simple modifications, I've created knowledge bases of plants, rocks and minerals, even famous people. Kids seem to especially enjoy the program. And as an educator, I can attest to its value as a tool for learning.

Here's the source code for one variant https://www.animalgame.com/play/misc/source.php:

100 &"PLAY 'GUESS THE ANIMAL' WITH RSTS
150 &"THINK OF AN ANIMAL AND THE COMPUTER WILL TRY TO GUESS IT...":&
500 DIM A$(200)
525 F$="ANIMAL.GME"
:ON ERROR GOTO 700
550 OPEN F$ FOR INPUT AS FILE 1%
:INPUT #1%,N%
:INPUT #1,A$(I%) FOR I%=1%TO N%
:CLOSE 1%
:A$(0%)=NUM$(N%)
:ON ERROR GOTO 0
:GOTO 1300
700 ON ERROR GOTO 1050
:F$="$"+F$
:RESUME 550
1050 READ A$(I%) FOR I%=0% TO 3%
1100 DATA "4","\QDOES IT SWIM\Y2\N3\","\AFISH","\ABIRD"
1300 INPUT "ARE YOU THINKING OF AN ANIMAL";Z9$
:GOTO 1350 IF LEFT(Z9$,1%)="Y"
:GOTO 4000 IF LEFT(Z9$,1%)="N"
1310 IF Z9$="SAVE" THEN
OPEN "ANIMAL.GME" FOR OUTPUT AS FILE 1%
:PRINT #1%,A$(I%) FOR I%=0% TO VAL(A$(0%))
:PRINT #1,CHR$(26%)
:CLOSE 1%
:GOTO 1300
1320 IF Z9$="LIST" THEN
PRINT "ANIMALS I ALREADY KNOW ARE:"
:PRINT RIGHT(A$(I%),3%), IF INSTR(1%,A$(I%),"\A") FOR I%=1% TO 200%
:PRINT
:GOTO 1300
1330 &"PLEASE RESPOND 'YES', 'NO', 'SAVE', OR 'LIST'":GOTO1300
1350 K%=1%
1400 K%=FNA%(A$(K%))
:GOTO 3000 IF LEN(A$(K%))=0%
:GOTO 1400 IF LEFT(A$(K%),2%)="\Q"
:PRINT "IS IT A "RIGHT(A$(K%),3%);
:INPUT Z7$
:Z7$=LEFT(Z7$,1%)
1450 IF Z7$="Y" THEN
PRINT "WHY NOT TRY ANOTHER ANIMAL"
:GOTO 1300
2000 INPUT "THE ANIMAL YOU WERE THINKING OF WAS A ";Z9$
2050 PRINT "PLEASE TYPE IN A QUESTION THAT WOULD DISTINGUISH A "
Z9$ " FROM A "RIGHT(A$(K%),3%)
:INPUT Z8$
2100 PRINT "FOR A "Z9$" THE ANSWER WOULD BE";
:INPUT Z7$
:Z7$=LEFT(Z7$,1%)
:IF Z7$="Y" THEN Z6$="N"
ELSE IF Z7$="N" THEN Z6$="Y"
ELSE PRINT "PLEASE ANSWER 'YES' OR 'NO'"
:GOTO 2100
2200 Z1%=VAL(A$(0%))
:A$(0%)=NUM$(Z1%+2%)
:A$(Z1%)=A$(K%)
:A$(Z1%+1%)='\A'+Z9$
:A$(K%)="\Q"+Z8$+"\"+Z7$+NUM$(Z1%+1%)+"\"+Z6$+NUM$(Z1%)+"\"
2300 GOTO 1300
3000 DEF FNA%(Q$)
:PRINT MID(Q$,3%,INSTR(3%,Q$,"\")-3%);
:INPUT Z9$
:Z9$=LEFT(Z9$,1%)
:Z9$="N" IF Z9$<>"Y"
:Z1%=INSTR(3%,Q$,"\"+Z9$)+2%
:Z2%=INSTR(Z1%,Q$,"\")
:FNA%=VAL(MID(Q$,Z1%,Z2%-Z1%))
:FNEND
4000 &:&"O.K. SEE YOU LATER. HOPE YOU HAD FUN PLAYING!!":&:&
9999 END